


CORSO DI COSTRUZIONI MECCANICHE II  
ESERCITAZIONE SULL'IMPIEGO DEL MEF

**ANALISI DI RECIPIENTE CILINDRICO IN PRESSIONE**

E' dato il recipiente in pressione, cilindrico con fondi emisferici, mostrato in Figura. Si metta a punto un modello ad EF e si determini l'andamento delle componenti di tensione nella zona cilindrica, confrontando i risultati ottenuti con quelli forniti dalla teoria dei corpi cilindrici soggetti a pressione interna.



**DATI**

$H := 3000 \cdot \text{mm}$

$R_{INT} := 1000 \cdot \text{mm}$

$SP := 300 \cdot \text{mm}$

$P_{INT} := 10 \cdot \text{MPa}$       Pressione interna


## CALCOLO TENSIONI

Dalla teoria dei corpi cilindrici chiusi soggetti a pressione interna si ha:

$$p_i := \text{PINT} \quad R_i := \text{RINT} \quad R_e := \text{RINT} + \text{SP} \quad \text{xx}(r) := 0$$

$$\sigma_a := p_i \cdot \frac{\pi \cdot R_i^2}{\pi \cdot (R_e^2 - R_i^2)}$$

$$\sigma_r(r) := \frac{p_i \cdot R_i^2}{R_e^2 - R_i^2} - \frac{R_e^2 \cdot R_i^2}{R_e^2 - R_i^2} \cdot \frac{p_i}{r^2} \quad \sigma_\theta(r) := \frac{p_i \cdot R_i^2}{R_e^2 - R_i^2} + \frac{R_e^2 \cdot R_i^2}{R_e^2 - R_i^2} \cdot \frac{p_i}{r^2}$$



$$\sigma_r(R_i) = -10 \text{ MPa} \quad \sigma_r(R_e) = -1.863 \times 10^{-15} \text{ MPa}$$

$$\sigma_\theta(R_i) = 38.986 \text{ MPa} \quad \sigma_\theta(R_e) = 28.986 \text{ MPa}$$

$$\sigma_a = 14.493 \text{ MPa}$$

## MODELLO FEM

Il modello FEM può essere costruito, data la semplicità della geometria, sia con procedura "Top-Down" che "Bottom-Up".

In entrambi i casi, il modello rappresenta la metà di una sezione passante per l'asse di simmetria.


Nell'approccio "Top-Down" (App.1) vengono generate 3 primitive (i 2 fondi ed il fasciame cilindrico), mentre nell'approccio "Bottom-Up" (App. 2) si generano i "Keypoints", le linee e le aree.

nel primo caso, le tre aree ottenute vengono unite tra loro con un'operazione di "Glueing", mentre nel secondo caso questo non è necessario, dato che le tre aree vengono direttamente create con un lato in comune

Le tre aree sono infine suddivise in elementi 42 quadrilateri con una mesh di tipo "mapped".


I vincoli relativi all'asse di simmetria sono imposti automaticamente, mentre è necessario un vincolo assiale per evitare la labilità. Quest'ultimo è imposto su di un "keypoint" appositamente selezionato.

Il carico di pressione interna è applicato alle linee che delimitano la superficie interna del modello, anch'esse preliminarmente selezionate attraverso la posizione del loro punto medio. Il modello ottenuto è mostrato nella Figura.





I risultati forniti dal modello sono riportati nel seguito. Tra di essi compare anche la rappresentazione dell'andamento di grandezze specificate (nel caso presente, le componenti di tensione radiale (SR), circonferenziale (ST) e assiale (SA)) lungo un percorso che si sviluppa all'interno del modello (nel caso presente, un segmento ce attraversa lo spessore in direzione radiale). Tale rappresentazione è ottenuta con i comandi PATH, PPATH, PDEF e PLPATH

Deformata (si noti l'espansione assiale e radiale):



Mappa delle tensioni radiali (SX), Assiali (SY) e circonferenziali (SZ).





1

NODAL SOLUTION

STEP=1  
SUB =1  
TIME=1  
SZ (AVG)  
RSYS=0  
DMX =.204842  
SMN =11.247  
SMX =39.683


ANSYS

MAY 17 2006  
12:33:57



Andamento delle tre componenti di tensione lungo la coordinata radiale, a metà altezza del recipiente. Si notino:

- la costanza della tensione radiale
- l'andamento non lineare delle tensioni radiale e circonferenziale
- la concordanza dei valori trovati con le previsioni del modello analitico



## APPENDICE 1

```
C*** -----
C***
C*** ESERCITAZIONE SULL'IMPIEGO DI ELEMENTI ASSIALSIMMETRICI
C***
C*** RECIPIENTE IN PRESSIONE
C***
c*** METODO TOP-DOWN
C***
FINISH
/CLEAR
/PREP7
C***
C*** DEFINIZIONE PARAMETRI
C***
H=3000
RINT=1000
SP=300
C***
C*** CREAZIONE MODELLO
C***
CYL4,0,H,RINT,0,RINT+SP,90 ! Fondo superiore
BLC4,RINT,0,SP,H ! Fasciame cilindrico
CYL4,0,0,RINT,0,RINT+SP,-90 ! Fondo inferiore
C***
C*** OPERAZIONE BOOLEANA
C***
AGLUE,ALL ! Unione (Glue) delle aree
C***
c*** MESH
C***
ET,1,42,,,1 ! Tipo di elemento
MSHAPE,0 ! Elementi quadrilateri
MSHKEY,1 ! Mesh "mapped"
AESIZE,ALL,SP/10 ! Dimensioni elementi
AMESH,ALL ! Mesh automatico delle aree
C***
C*** MATERIALE
C***
MP,EX,1,210000
MP,PRXY,1,0.3
C***
C*** VINCOLO ASSIALE SU DI UN KEYPOINT
C***
KSEL,,LOC,X,-1,0.0001 !Selezione keypoint
KSEL,R,LOC,Y,-(RINT+SP)-0.001,-(RINT+SP)+0.001
DK,ALL,UY,0 ! Vincolo
KSEL,ALL
C***
C*** PRESSIONE SULLA SUPERFICIE INTERNA
C***
LSEL,,LOC,Y,-RINT*SQRT(2)/2,H+RINT*SQRT(2)/2 ! Selezione linee
LSEL,R,LOC,X,0,RINT+0.001
SFL,ALL,PRESS,10 ! Pressione
```

LSEL,ALL

C\*\*\*

C\*\*\* SOLUZIONE

C\*\*\*

/SOLU

SOLVE

C\*\*\*

C\*\*\* POST-PROCESSING

C\*\*\*

/POST1

SET

PLDISP,1 ! Deformata

\*ASK,LAB0,RETURN PER CONTINUARE

PLNSTR,SX ! Tensione radiale

\*ASK,LAB0,RETURN PER CONTINUARE

PLNSTR,SY ! Tensione assiale

\*ASK,LAB0,RETURN PER CONTINUARE

PLNSTR,SZ ! Tensione circonferenziale

\*ASK,LAB0,RETURN PER CONTINUARE

!

! Tensioni lungo il raggio

!

PATH,RAGGIO,2,,100 ! Definizione percorso per rappresentazione

PPATH,1,,RINT,H/2,0 ! Punto 1 percorso

PPATH,2,,RINT+SP,H/2,0 ! Punto 2 percorso

PDEF,SR,S,X ! variabile SR = tensione radiale

PDEF,ST,S,Z ! variabile ST = tensione circonferenziale

PDEF,SA,S,Y ! variabile SA = tensione assiale

PLPATH,SR,ST,SA ! Rappresentazione grafica

## APPENDICE 2

```
C*** -----
C***
C*** ESERCITAZIONE SULL'IMPIEGO DI ELEMENTI ASSIALSIMMETRICI
C***
C*** RECIPIENTE IN PRESSIONE
C***
c*** METODO bottom-up
C***
FINISH
/CLEAR
/PREP7
C***
C*** DEFINIZIONE PARAMETRI
C***
H=3000
RINT=1000
SP=300

C***
C*** CREAZIONE MODELLO
C***
K,1,0,H+RINT          !***** FONDO SUPERIORE
K,2,0,H+RINT+SP      !Keypoints
K,3,RINT+SP,H
K,4,RINT,H
K,5,0,H
L,1,2                 !Linee
LARC,2,3,5,RINT+SP
L,3,4
LARC,4,1,5,RINT
AL,1,2,3,4           !Area
K,6,RINT,            !***** FASCIAME CILINDRICO
K,7,RINT+SP         !Keypoints
L,3,7                !Linee
L,7,6
L,6,4
AL,3,5,6,7          !Area
K,8,0,-(RINT+SP)    !***** FONDO INFERIORE
K,9,0,-RINT         !Keypoints
K,10,0
LARC,7,8,10,RINT+SP !Linee
L,8,9
LARC,9,6,10,RINT
AL,6,8,9,10         !Area

C***
c*** MESH
C***
ET,1,42,,1          ! Tipo di elemento
MSHAPE,0            ! Elementi quadrilateri
MSHKEY,1            ! Mesh "mapped"
AESIZE,ALL,SP/10    ! Dimensioni elementi
```

```

AMESH,ALL          ! Mesh automatico delle aree
C***
C*** MATERIALE
C***
MP,EX,1,210000
MP,PRXY,1,0.3
C***
C*** VINCOLO ASSIALE SU DI UN KEYPOINT
C***
KSEL,,LOC,X,-1,0.0001 !Selezione keypoint
KSEL,R,LOC,Y,-(RINT+SP)-0.001,-(RINT+SP)+0.001
DK,ALL,UY,0          ! Vincolo
KSEL,ALL
C***
C*** PRESSIONE SULLA SUPERFICIE INTERNA
C***
LSEL,,LOC,Y,-RINT*SQRT(2)/2,H+RINT*SQRT(2)/2      ! Selezione linee
LSEL,R,LOC,X,0,RINT+0.001
SFL,ALL,PRESS,10          ! Pressione
LSEL,ALL

C***
C*** SOLUZIONE
C***
/SOLU
SOLVE

C***
C*** POST-PROCESSING
C***
/POST1
SET
PLDISP,1                ! Deformata
*ASK,LAB0,RETURN PER CONTINUARE
PLNSTR,SX                ! Tensione radiale
*ASK,LAB0,RETURN PER CONTINUARE
PLNSTR,SY                ! Tensione assiale
*ASK,LAB0,RETURN PER CONTINUARE
PLNSTR,SZ                ! Tensione circonferenziale
*ASK,LAB0,RETURN PER CONTINUARE
!
! Tensioni lungo il raggio
!
PATH,RAGGIO,2,,100      ! Definizione percorso per rappresentazione
PPATH,1,,RINT,H/2,0     ! Punto 1 percorso
PPATH,2,,RINT+SP,H/2,0  ! Punto 2 percorso
PDEF,SR,S,X             ! variabile SR = tensione radiale
PDEF,ST,S,Z             ! variabile ST = tensione circonferenziale
PDEF,SA,S,Y             ! variabile SA = tensione assiale
PLPATH,SR,ST,SA         ! Rappresentazione grafica

```